
graph twoway bar — Two-way bar plots

Description Quick start Menu Syntax
Options Remarks and examples References Also see

Description
twoway bar displays numeric (𝑦,𝑥) data as bars. twoway bar is useful for drawing bar plots of time-

series data or other equally spaced data and is useful as a programming tool. For finely spaced data, also

see [G-2] graph twoway spike.

Also see [G-2] graph bar for traditional bar charts and [G-2] graph twoway histogram for histograms.

Quick start
Bar graph

twoway bar y x

A horizontal bar graph

twoway bar y x, horizontal

Bar graph with bars 0.8 times the default width

twoway bar y x, barwidth(.8)

Bars that extend from 0 when the range of y does not include 0

twoway bar y x, base(20)

Add the title “My Title” to the graph

twoway bar y x, title(”My Title”)

Same as above, but add a title “X variable” to the 𝑥 axis

twoway bar y x, title(”My Title”) xtitle(”X variable”)

Same as above, but with separate graph areas for each value of catvar
twoway bar y x, xtitle(”X variable”) by(catvar, title(”My Title”))

Menu
Graphics > Two-way graph (scatter, line, etc.)

https://www.stata.com/manuals/g-2graphtwowayspike.pdf#g-2graphtwowayspike
https://www.stata.com/manuals/g-2graphbar.pdf#g-2graphbar
https://www.stata.com/manuals/g-2graphtwowayhistogram.pdf#g-2graphtwowayhistogram


graph twoway bar — Two-way bar plots 2

Syntax
twoway bar yvar xvar [ if ] [ in ] [ , options ]

options Description

vertical vertical bar plot; the default

horizontal horizontal bar plot

base(#) value to drop to; default is 0

barwidth(#) width of bar in xvar units

barlook options change look of bars

colorvar options change color of bars based on values of a variable

marker label options add marker labels; change look or position

axis choice options associate plot with alternative axis

twoway options titles, legends, axes, added lines and text, by, regions, name,
aspect ratio, etc.

Options base() and barwidth() are rightmost, and vertical and horizontal are unique; see

[G-4] Concept: repeated options.

Options
vertical and horizontal specify either a vertical or a horizontal bar plot. vertical is the default. If

horizontal is specified, the values recorded in yvar are treated as 𝑥 values, and the values recorded

in xvar are treated as 𝑦 values. That is, to make horizontal plots, do not switch the order of the two

variables specified.

In the vertical case, bars are drawn at the specified xvar values and extend up or down from 0

according to the corresponding yvar values. If 0 is not in the range of the 𝑦 axis, bars extend up or

down to the 𝑥 axis.

In the horizontal case, bars are drawn at the specified xvar values and extend left or right from 0

according to the corresponding yvar values. If 0 is not in the range of the 𝑥 axis, bars extend left or

right to the 𝑦 axis.

base(#) specifies the value from which the bar should extend. The default is base(0) when 0 falls

between the minimum and maximum of yvar. Otherwise, the default base is the value of yvar closest

to 0.

barwidth(#) specifies the width of the bar in xvar units. The default is width(1). When a bar is

plotted, it is centered at 𝑥, so half the width extends below 𝑥 and half above.

barlook options set the look of the bars. The most important of these options is

color(colorstyle), which specifies the color and opacity of the bars; see [G-4] colorstyle for a list of
color choices. See [G-3] barlook options for information on the other barlook options.

colorvar options specify that the color of the bars be determined by the levels of the numeric variable

colorvar; see [G-3] colorvar options.

marker label options specify labels to appear above the bars. See [G-3] marker label options for a

description of marker labels and the options that control them.

https://www.stata.com/manuals/u11.pdf#u11.1.3ifexp
https://www.stata.com/manuals/u11.pdf#u11.1.4inrange
https://www.stata.com/manuals/g-3barlook_options.pdf#g-3barlook_options
https://www.stata.com/manuals/g-3colorvar_options.pdf#g-3colorvar_options
https://www.stata.com/manuals/g-3marker_label_options.pdf#g-3marker_label_options
https://www.stata.com/manuals/g-3axis_choice_options.pdf#g-3axis_choice_options
https://www.stata.com/manuals/g-3twoway_options.pdf#g-3twoway_options
https://www.stata.com/manuals/g-4conceptrepeatedoptions.pdf#g-4Conceptrepeatedoptions
https://www.stata.com/manuals/g-4colorstyle.pdf#g-4colorstyle
https://www.stata.com/manuals/g-3barlook_options.pdf#g-3barlook_options
https://www.stata.com/manuals/g-3colorvar_options.pdf#g-3colorvar_options
https://www.stata.com/manuals/g-3marker_label_options.pdf#g-3marker_label_options

axis choice options associate the plot with a particular 𝑦 or 𝑥 axis on the graph; see

[G-3] axis choice options.

twoway options are a set of common options supported by all twoway graphs. These options allow you

to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create

graphs over by() groups, and change some advanced settings. See [G-3] twoway options.

Remarks and examples
Remarks are presented under the following headings:

Typical use
Advanced use: Overlaying
Advanced use: Population pyramid
Cautions

Typical use
We have daily data recording the values for the S&P 500 in 2001:

. use https://www.stata-press.com/data/r19/sp500
(S&P 500)
. list date close change in 1/5

date close change

1. 02jan2001 1283.27 .
2. 03jan2001 1347.56 64.29004
3. 04jan2001 1333.34 -14.22009
4. 05jan2001 1298.35 -34.98999
5. 08jan2001 1295.86 -2.48999

We will use the first 57 observations from these data:

. twoway bar change date in 1/57

https://www.stata.com/manuals/g-3axis_choice_options.pdf#g-3axis_choice_options
https://www.stata.com/manuals/g-3twoway_options.pdf#g-3twoway_options

We get a different visual effect if we reduce the width of the bars from 1 day to .6 days:

. twoway bar change date in 1/57, barw(.6)

Advanced use: Overlaying
The useful thing about twoway bar is that it can be combined with other twoway plottypes (see

[G-2] graph twoway):

. twoway line close date || bar change date || in 1/52

https://www.stata.com/manuals/g-2graphtwoway.pdf#g-2graphtwoway

We can improve this graph by typing

. twoway
line close date, yaxis(1)

||
bar change date, yaxis(2)

||
in 1/52,

ysca(axis(1) r(1000 1400)) ylab(1200(50)1400, axis(1))
ysca(axis(2) r(-50 300)) ylab(-50 0 50, axis(2))

ytick(-50(25)50, axis(2) grid)
legend(off)
xtitle(”Date”)
title(”S&P 500”)
subtitle(”January to March 2001”)
note(”Source: Yahoo!Finance and Commodity Systems, Inc.”)
yline(1150, axis(1) lstyle(foreground))

Notice the use of

yline(1150, axis(1) lstyle(foreground))

The 1150 put the horizontal line at 𝑦 = 1150; axis(1) stated that 𝑦 should be interpreted according

to the left 𝑦 axis; and lstyle(foreground) specified that the line be drawn in the foreground style.

Advanced use: Population pyramid
We have the following aggregate data from the US 2000 Census recording total population by age and

sex. From this, we produce a population pyramid:

. use https://www.stata-press.com/data/r19/pop2000, clear
(2000 US Census population by age and sex)
. list agegrp maletotal femtotal

agegrp maletotal femtotal

1. Under 5 9,810,733 9,365,065
2. 5 to 9 10,523,277 10,026,228
3. 10 to 14 10,520,197 10,007,875
4. 15 to 19 10,391,004 9,828,886
5. 20 to 24 9,687,814 9,276,187

6. 25 to 29 9,798,760 9,582,576
7. 30 to 34 10,321,769 10,188,619
8. 35 to 39 11,318,696 11,387,968
9. 40 to 44 11,129,102 11,312,761

10. 45 to 49 9,889,506 10,202,898

11. 50 to 54 8,607,724 8,977,824
12. 55 to 59 6,508,729 6,960,508
13. 60 to 64 5,136,627 5,668,820
14. 65 to 69 4,400,362 5,133,183
15. 70 to 74 3,902,912 4,954,529

16. 75 to 79 3,044,456 4,371,357
17. 80 to 84 1,834,897 3,110,470

. replace maletotal = -maletotal/1e+6
(17 real changes made)
. replace femtotal = femtotal/1e+6
(17 real changes made)
. twoway

bar maletotal agegrp, horizontal xvarlab(Male)
||

bar femtotal agegrp, horizontal xvarlab(Female)
||
, ylabel(1(1)17, valuelabel labsize(*.8))

xtitle(”Population in millions”) ytitle(””)
xlabel(-10 ”10” -7.5 ”7.5” -5 ”5” -2.5 ”2.5” 2.5 5 7.5 10)
legend(label(1 Male) label(2 Female))
title(”US male and female population by age”)
subtitle(”Year 2000”)
note(”Source: US Census Bureau, Census 2000, Tables 1, 2, and 3”, span)

Under 5
5 to 9

10 to 14
15 to 19
20 to 24
25 to 29

35 to 39

60 to 64
65 to 69
70 to 74
75 to 79
80 to 84

Population in millions

 Female

Source: US Census Bureau, Census 2000, Tables 1, 2, and 3

Year 2000

US male and female population by age

At its heart, the above graph is simple: we turned the bars sideways and changed the male total to be

negative. Our first attempt at the above was simply

. use https://www.stata-press.com/data/r19/pop2000, clear
(2000 US Census population by age and sex)
. replace maletotal = -maletotal
(17 real changes made)
. twoway bar maletotal agegrp, horizontal ||

bar femtotal agegrp, horizontal

From there, we divided the population totals by 1 million and added options.

xlabel(-10 ”10” -7.5 ”7.5” -5 ”5” -2.5 ”2.5” 2.5 5 7.5 10)was a clever way to disguise that
the bars for males extended in the negative direction. We said to label the values −10, −7.5, −5, −2.5,

2.5, 5, 7.5, and 10, but then we substituted text for the negative numbers to make it appear that they were

positive. See [G-3] axis label options.

Using the span suboption to note() aligned the text on the left side of the graph rather than on the

plot region. See [G-3] textbox options.

https://www.stata.com/manuals/g-3axis_label_options.pdf#g-3axis_label_options
https://www.stata.com/manuals/g-3textbox_options.pdf#g-3textbox_options

For another rendition of the pyramid, we tried

. use https://www.stata-press.com/data/r19/pop2000, clear
(2000 US Census population by age and sex)
. replace maletotal = -maletotal/1e+6
(17 real changes made)
. replace femtotal = femtotal/1e+6
(17 real changes made)
. generate zero = 0
. twoway

bar maletotal agegrp, horizontal xvarlab(Male)
||

bar femtotal agegrp, horizontal xvarlab(Female)
||

sc agegrp zero , mlabel(agegrp) mlabcolor(black) msymbol(i)
||

, xtitle(”Population in millions”) ytitle(””)
ysca(noline) ylabel(none) (note 1)
xsca(noline titlegap(-3.5)) (note 2)
xlabel(-12 ”12” -10 ”10” -8 ”8” -6 ”6” -4 ”4” 4(2)12, tlength(0)

gmin gmax)
legend(label(1 Male) label(2 Female)) legend(order(1 2))
title(”US male and female population by age, 2000”)
note(”Source: US Census Bureau, Census 2000, Tables 1, 2, and 3”)

Source: US Census Bureau, Census 2000, Tables 1, 2, and 3

US male and female population by age, 2000

In the above rendition, we moved the labels from the 𝑥 axis to inside the bars by overlaying a

scatter on top of the bars. The points of the scatter we plotted at 𝑦 = agegrp and 𝑥 = 0, and

rather than showing the markers, we displayed marker labels containing the desired labelings. See

[G-3] marker label options.

We also played the following tricks:

1. ysca(noline) suppressed drawing the 𝑦 axis—see [G-3] axis scale options—and

ylabel(none) suppressed labeling it—see [G-3] axis label options.

2. xsca(noline titlegap(-3.5)) suppressed drawing the 𝑥 axis and moved the 𝑥-axis title up
to be in between its labels; see [G-3] axis scale options.

https://www.stata.com/manuals/g-3marker_label_options.pdf#g-3marker_label_options
https://www.stata.com/manuals/g-3axis_scale_options.pdf#g-3axis_scale_options
https://www.stata.com/manuals/g-3axis_label_options.pdf#g-3axis_label_options
https://www.stata.com/manuals/g-3axis_scale_options.pdf#g-3axis_scale_options

Cautions
You must extend the scale of the axis, if that is necessary. Consider using twoway bar to produce

a histogram (ignoring the better alternative of using twoway histogram; see [G-2] graph twoway his-

togram). Assume that you have already aggregated data of the form

x frequency

1 400
2 800
3 3,000
4 1,800
5 1,100

which you enter into Stata to make variables x and frequency. You type

. twoway bar frequency x

to make a histogram-style bar chart. The 𝑦 axis will be scaled to go between 400 and 3,000 (labeled at

500, 1,000, . . . , 3,000), and the shortest bar will have zero height. You need to type

. twoway bar frequency x, ysca(r(0))

References
Cox, N. J., and T. P. Morris. 2024. Speaking Stata: The joy of sets: Graphical alternatives to Euler and Venn diagrams.

Stata Journal 24: 329–361.

Jann, B. 2015. Stata tip 122: Variable bar widths in two-way graphs. Stata Journal 15: 316–318.

Also see
[G-2] graph twoway scatter — Two-way scatterplots

[G-2] graph twoway dot — Two-way dot plots

[G-2] graph twoway dropline — Two-way dropped-line plots

[G-2] graph twoway histogram — Histogram plots

[G-2] graph twoway spike — Two-way spike plots

[G-2] graph bar — Bar charts

Stata, Stata Press, Mata, NetCourse, and NetCourseNow are registered trademarks of StataCorp
LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Or-
ganization of the United Nations. StataNow is a trademark of StataCorp LLC. Other brand and
product names are registered trademarks or trademarks of their respective companies. Copyright
© 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

®

For suggested citations, see the FAQ on citing Stata documentation.

https://www.stata.com/manuals/g-2graphtwowayhistogram.pdf#g-2graphtwowayhistogram
https://www.stata.com/manuals/g-2graphtwowayhistogram.pdf#g-2graphtwowayhistogram
https://doi.org/10.1177/1536867X241258010
https://www.stata-journal.com/article.html?article=gr0063
https://www.stata.com/manuals/g-2graphtwowayscatter.pdf#g-2graphtwowayscatter
https://www.stata.com/manuals/g-2graphtwowaydot.pdf#g-2graphtwowaydot
https://www.stata.com/manuals/g-2graphtwowaydropline.pdf#g-2graphtwowaydropline
https://www.stata.com/manuals/g-2graphtwowayhistogram.pdf#g-2graphtwowayhistogram
https://www.stata.com/manuals/g-2graphtwowayspike.pdf#g-2graphtwowayspike
https://www.stata.com/manuals/g-2graphbar.pdf#g-2graphbar
https://www.stata.com/support/faqs/resources/citing-software-documentation-faqs/

